
This Lecture

You know you have a distributed system when the
crash of a computer you’ve never heard of stops
you from getting any work done.–Leslie Lamport

Using Promela channels for modeling distributed systems

Modeling Distributed Systems

Distributed systems consist of

I nodes,

I interacting via communication channels,

I protocols dictate how nodes communicate with each other.

Distributed systems are very complex.

Models of distributed systems abstract away from details of
networks/protocols/nodes.

In Promela:

I nodes modeled by Promela processes

I communication channels modeled by Promela channels

I protocols modeled by algorithm distributed over processes

Channels in Promela

In Promela, channels are first class citizens.

Data type chan with two operations for sending and receiving

A variable of channel type is declared by initializer:

chan name = [capacity] of {type1, ..., typen}

name name of channel variable
capacity non-negative integer constant
type i Promela data types

Example:
chan ch = [2] of { mtype, byte, bool }

Meaning of Channels

chan name = [capacity] of {type1, ..., typen}

Creates channel, stored in variable name

Messages communicated via channel are n-tuples∈ type1 × . . .× typen

Can buffer up to capacity messages, if capacity ≥ 1
⇒ “buffered channel”

The channel has no buffer if capacity = 0
⇒ “rendezvous channel”

Meaning of Channels

Example:

chan ch = [2] of { mtype, byte, bool }

Creates channel, stored in variable ch

Messages communicated via ch are 3-tuples ∈ mtype × byte × bool

Given, e.g., mtype = {red, yellow, green},
an example message on ch can be: green, 20, false

ch is a buffered channel, buffering up to 2 messages

Sending and Receiving

send statement has the form:

name ! expr1, ... , exprn
I name: channel variable
I expr1, ... , exprn: sequence of expressions,

where number and types match name’s type
I sends values of expr1, ... , exprn as one message
I example: ch ! green, i+20, false

receive statement has the form:

name ? var1, ... , varn
I name: channel variable
I var1, ... , varn: sequence of variables,

where number and types match name’s type
I assigns values of message to var1, ... , varn
I example: ch ? color, time, flash

Client-Server

chan request = [0] of { byte };

active proctype Client0 () {

request ! 0

}

active proctype Client1 () {

request ! 1

}

...

Client0 and Client1 send messages 0 resp. 1 to channel request

Order of sending is nondeterministic

Client-Server

chan request = [0] of { byte };

...

active proctype Server () {

byte num;

do
:: request ? num;

print f ("serving client %d\n", num)

od
}

Server loops on

I receiving first message from request, storing value in num

I printing

Executability of receive Statement (non-buffered)

request ? num

executable only when another process offers send on channel request

⇒ receive statement frequently used as guard in if/do-statements

do
:: request ? num ->

print f ("serving client %d\n", num)

od

(“->” equivalent to “;”)

Rendezvous Channels

chan ch = [0] of { byte, byte };

/* global only to make visible in SpinSpider */

byte hour , minute;

active proctype Sender () {

print f ("ready\n");
ch ! 11, 45;

print f ("Sent\n")
}

active proctype Receiver () {

print f ("steady\n");
ch ? hour , minute;

print f ("Received\n")
}

Which interleavings can occur? ⇒ ask SpinSpider

Rendezvous are Synchronous

On a rendezvous channel:

Transfer of message from sender to receiver is synchronous,
i.e., one single operation.

Sender Receiver
...

...
(11,45) −→ (hour,minute)

...
...

FMSD: Modeling Distributed Systems /GU 180914 14 / 36



Rendezvous are Synchronous

Either:

1. Location counter of sender process at send (“!”):
“offer to engage in rendezvous”

2. Location counter of receiver process at receive (“?”):
“rendezvous can be accepted”

or the other way round:

1. Location counter of receiver process at receive (“?”):
“offer to engage in rendezvous”

2. Location counter of sender process at send (“!”):
“rendezvous can be accepted”

In both case, the next step is:

Location counter of both processes is incremented at once.

Only place where Promela processes execute synchronously

Reconsider Client Server

chan request = [0] of { byte };

active proctype Server () {

byte num;

do :: request ? num ->

print f ("serving client %d\n", num)

od
}

active proctype Client0 () {

request ! 0

}

active proctype Client1 () {

request ! 1

}

So far no reply to clients

Reply Channels

chan request = [0] of { byte };

chan ack = [0] of { bool };

active proctype Server () {

byte num;

do :: request ? num ->

print f ("serving client %d\n", num);

ack ! true
od

}

active proctype Client0 () {

request ! 0; ack ? _; print f ("acknowledged\n")
}

active proctype Client1 () {

request ! 1; ack ? _; print f ("acknowledged\n")
}

(Anonymous variable “_”: data from channel no stored anywhere)

Reply Channels - Single Server

mtype = { nice , rude };

chan request = [0] of { mtype };

chan reply = [0] of { mtype };

active proctype Server () {

mtype msg;

do :: request ? msg; reply ! msg

od
}

active proctype NiceClient () {

mtype msg;

request ! nice; reply ? msg;

assert(msg == nice) Is the assertion valid? Ask Spin.
}

active proctype RudeClient () {

mtype msg;

request ! rude; reply ? msg

}

Several Servers

More realistic with several servers:

active [2] proctype Server () {

mtype msg;

do :: request ? msg; reply ! msg

od
}

active proctype NiceClient () {

mtype msg;

request ! nice; reply ? msg;

assert(msg == nice) And here? Analyse with Spin.
}

active proctype RudeClient () {

mtype msg;

request ! rude; reply ? msg

}

Sending Channels via Channels

To fix the protocol:

clients declare local reply channel + send it to server

Sending Channels via Channels

mtype = { nice , rude };

chan request = [0] of { mtype, chan };

active [2] proctype Server () {

mtype msg; chan ch;

do :: request ? msg , ch;

ch ! msg

od
}

active proctype NiceClient () {

chan reply = [0] of { mtype }; mtype msg;

request ! nice , reply; reply ? msg;

assert ( msg == nice )

}

active proctype RudeClient () {

chan reply = [0] of { mtype }; mtype msg;

request ! rude , reply; reply ? msg

}
verify with Spin

Scope of Channels

Global channel

I All processes can send and/or receive messages

Local channel

I Can model ‘private’ communication & security issues
I Example:

Local channel can be passed through a global channel

Sending Process IDs

Used fixed constants used for identification (here nice, rude)

I inflexible

I doesn’t scale

Alternative:
Processes send their own, unique process ID, _pid, as part of message

Experiment with rendezvous3.pml

Example, clients code:

chan reply = [0] of { byte, byte };

request ! pid, reply;

reply ? serverID , serversClient;

assert ( serversClient == pid )

Limitations of Rendezvous Channels

I Rendezvous too restrictive for many applications

I Servers and clients block each other too much

I Difficult to manage uneven workload
(online shop: dozens of webservers serve thousands of clients)

Buffered Channel

Buffered channels queue messages.
Requests/services no not immediately block clients/servers.

Example:
chan ch = [3] of { mtype, byte, bool }

Buffered Channels

Buffered channels, with capacity cap

I Can hold up to cap messages

I Are a FIFO (first-in-first-out) data structure:
always the ‘oldest’ message in channel is retrieved by a receive

I (Normal) receive statement reads and removes message

I Sending and receiving to/from buffered channels is asynchronous,
i.e. interleaved

Executability of Buffered Channel operations

Given channel ch, with capacity cap, currently containing n messages

receive statement ch ? msg

is executable iff ch is not empty, i.e., n > 0

send statement ch ! msg

is executable iff there is still ‘space’ in the message queue,
i.e., n < cap

A non-executable receive or send statement will block until it is
executable again

(With option -m, Spin has a different send semantics:
Attempt to send to full channel doesn’t block, but message gets lost.)

Checking Channel for Full/Empty

This can prevent unnecessary blocking:

Given channel ch:

full(ch) checks whether ch is full
nfull(ch) checks whether ch is not full
empty(ch) checks whether ch is empty
nempty(ch) checks whether ch is not empty

Illegal to negate those.
Avoid combining with else.

Copy Message without Removing

Assume ch to be a buffered channel.

ch ? color, time, flash

I Assigns values from the message to color, time, flash

I Removes message from ch

ch ? <color, time, flash>

I Assign values from the message to color, time, flash

I Leaves message in ch

Dispatching Messages

Recurring task: Dispatch action depending on message

mtype = {hi , bye};

chan ch = [0] of {mtype};

active proctype Server () {

mtype msg;

read:

ch ? msg;

do
:: msg == hi -> print f ("Hello .\n"); goto read

:: msg == bye -> print f ("See you.\n"); break
od

}

...

There is a better way!

Dispatching Messages

Recurring task: Dispatch action depending on message type.

mtype = {hi , bye};

chan ch = [0] of {mtype};

active proctype Server () {

do
:: ch ? hi -> print f ("Hello.\n")
:: ch ? bye -> print f ("See you.\n"); break

od
}

...

hi and bye are values, not variables!

Pattern Matching

Receive statement allows also non-variable expressions as arguments:

ch ? exp1, . . . , expn

I exp1, . . . , expn any(!) expressions of correct type
I Receive statement is executable, iff

1. either
I ch is buffered channel and not empty, or
I ch is rendezvous channel and some process ready to send to ch

2. message msg1, . . . ,msgn in channel ch matches exp1, . . . , expn

I msgi matches expi iff
I expi is a variable and msgi a value (of correct type)
I expi is not a variable and expi == msgi

Pattern Matching Examples

Assume

chan ch = [0] of { int , int };
int id = 5;

Does ch ? 0, id match message

I [0, 5] ? 4 [0, 7] ? 4 [1, 7] ? 8

I Value of id afterwards?

To match the value stored in a variable var use eval(var)

Does ch ? 0, eval(id) match message

I [0, 5] ? 4 [0, 7] ? 8 [1, 7] ? 8

I Value of id afterwards?

Dispatching Messages Revisited

Random receive ?? (for buffered channels)

I Executable if matching message exists in channel.

I If executed, first matching message removed from channel.

mtype = {hi , bye};

chan ch = [3] of {mtype};

active proctype Server () {

do
:: ch ?? bye -> print f ("See you.\n"); break
:: e l se -> print f ("Hello .\n")

od
}

...

Nicer Message Formatting

Promela provides an alternative, but equivalent syntax for

ch ! exp1, exp2, exp3

namely

ch ! exp1(exp2, exp3)

Increases readability for certain applications, e.g. protocol modelling:
ch!send(msg,id) vs. ch!send,msg,id

ch!ack(id) vs. ch!ack,id

And finally

Buffered channels are part of the state!

State space gets much bigger using buffered channels

Use with care (and with small buffers).

